



# Bài 4

## BNF và sơ đồ cú pháp

ONE LOVE. ONE FUTURE.

- **Siêu ngữ (metalanguage)** :Ngôn ngữ sử dụng các lệnh để mô tả ngôn ngữ khác
- **BNF (Backus Naur Form)** là dạng siêu cú pháp để mô tả các ngôn ngữ lập trình
- BNF được sử dụng rộng rãi để mô tả văn phạm của các ngôn ngữ lập trình, tập lệnh và các giao thức truyền thông.

- Ký pháp BNF là một tập các luật ,vế trái của mỗi luật là một cấu trúc cú pháp.
- Tên của cấu trúc cú pháp được gọi là ký hiệu không kết thúc.
- Các ký hiệu không kết thúc thường được bao trong cặp  $\langle \rangle$ .
- Các ký hiệu kết thúc thường được phân cách bằng cặp nháy đơn hoặc nháy kép

- Mỗi ký hiệu không kết thúc được định nghĩa bằng một hay nhiều luật.
- Các luật có dạng

$$N::=s$$

*(N là ký hiệu không kết thúc, s là một xâu gồm 0 hay nhiều ký hiệu kết thúc và không kết thúc. Các luật có chung vế trái được phân cách bằng | )*

## Ví dụ về BNF : văn phạm sản sinh các số thực

$\langle \text{số thực} \rangle ::= \langle \text{dấu} \rangle \langle \text{số tự nhiên} \rangle \mid$   
 $\langle \text{dấu} \rangle \langle \text{số tự nhiên} \rangle '.' \langle \text{dãy chữ số} \rangle \mid$   
 $\langle \text{dấu} \rangle '.' \langle \text{chữ số} \rangle \langle \text{dãy chữ số} \rangle \mid$   
 $\langle \text{số thực} \rangle 'e' \langle \text{số tự nhiên} \rangle$

$\langle \text{dấu} \rangle ::= \varepsilon \mid '+' \mid '-'$

$\langle \text{số tự nhiên} \rangle ::= '0' \mid \langle \text{chữ số khác } 0 \rangle \langle \text{dãy chữ số} \rangle$

$\langle \text{chữ số khác } 0 \rangle ::= '1' \mid '2' \mid '3' \mid '4' \mid '5' \mid '6' \mid '7' \mid '8' \mid '9'$

$\langle \text{dãy chữ số} \rangle ::= \varepsilon \mid \langle \text{chữ số} \rangle \langle \text{dãy chữ số} \rangle$

$\langle \text{chữ số} \rangle ::= '0' \mid '1' \mid '2' \mid '3' \mid '4' \mid '5' \mid '6' \mid '7' \mid '8' \mid '9'$

Văn phạm này được viết bằng BNF, một công cụ rất phổ biến để biểu diễn cú pháp ngôn ngữ lập trình

- EBNF (Extended BNF ) được phát triển từ ký pháp BNF. EBNF có ký pháp tương tự BNF nhưng được đơn giản hoá bằng cách sử dụng một số ký hiệu đặc biệt :

[] phần này là tùy chọn(có hoặc không)

{ } phần này có thể lặp lại một số lần tùy ý hoặc không xuất hiện lần nào (Nếu lặp lại m hay n lần , dùng n hay m là chỉ số trên hoặc dưới)

Không cần dùng " cho ký hiệu kết thúc

## Ví dụ

- Trong EBNF

$\langle \text{Lệnh if} \rangle ::= \text{IF } \langle \text{Biểu thức} \rangle \text{ THEN } \langle \text{Lệnh} \rangle [\text{ELSE } \langle \text{Lệnh} \rangle]$

- Trong BNF

$\langle \text{Lệnh if} \rangle ::= \text{'IF' } \langle \text{Biểu thức} \rangle \text{'THEN' } \langle \text{Lệnh} \rangle | \text{'IF' } \langle \text{Biểu thức} \rangle \text{'THEN' } \langle \text{Lệnh} \rangle \text{'ELSE' } \langle \text{Lệnh} \rangle$

## Ví dụ: Một đoạn văn phạm Python trên EBNF

```
compound_stmt: if_stmt | while_stmt | for_stmt | try_stmt | with_stmt  
| funcdef | classdef | decorated | async_stmt  
async_stmt: 'async' (funcdef | with_stmt | for_stmt)  
if_stmt: 'if' test ':' suite ('elif' test ':' suite)* ['else' ':' suite]  
while_stmt: 'while' test ':' suite ['else' ':' suite]  
for_stmt: 'for' exprlist 'in' testlist ':' suite ['else' ':' suite]  
try_stmt: ('try' ':' suite  
          ((except_clause ':' suite)+  
           ['else' ':' suite]  
           ['finally' ':' suite] |  
           'finally' ':' suite))  
with_stmt: 'with' with_item (',' with_item)* ':' suite  
with_item: test ['as' expr]
```



# Văn phạm KPL viết bằng BNF

- 01) `<Prog> ::= KW_PROGRAM TK_IDENT SB_SEMICOLON <Block> SB_PERIOD`
- 02) `<Block> ::= KW_CONST <ConstDecl> <ConstDecls> <Block2>`
- 03) `<Block> ::= <Block2>`
- 04) `<Block2> ::= KW_TYPE <TypeDecl> <TypeDecls> <Block3>`
- 05) `<Block2> ::= <Block3>`
- 06) `<Block3> ::= KW_VAR <VarDecl> <VarDecls><Block4>`
- 07) `<Block3> ::= <Block4>`
- 08) `<Block4> ::= <SubDecls><Block5>`
- 09) `<Block4> ::= <Block5>`
- 10) `<Block5> ::= KW_BEGIN <Statements> KW_END`
- 11) `<ConstDecls> ::= <ConstDecl> <ConstDecls>`
- 12) `<ConstDecls> ::= ε`
- 13) `<ConstDecl> ::= TK_IDENT SB_EQUAL <Constant> SB_SEMICOLON`
- 14) `<TypeDecls> ::= <TypeDecl> <TypeDecls>`
- 15) `<TypeDecls> ::= ε`
- 16) `<TypeDecl> ::= TK_IDENT SB_EQUAL <Type> SB_SEMICOLON`

# Văn phạm KPL viết bằng BNF

- 17)  $\langle \text{VarDecls} \rangle ::= \langle \text{VarDecl} \rangle \langle \text{VarDecls} \rangle$
- 18)  $\langle \text{VarDecls} \rangle ::= \epsilon$
  
- 19)  $\langle \text{VarDecl} \rangle ::= \text{TK\_IDENT SB\_COLON} \langle \text{Type} \rangle \text{SB\_SEMICOLON}$
  
- 20)  $\langle \text{SubDecls} \rangle ::= \langle \text{FunDecl} \rangle \langle \text{SubDecls} \rangle$
- 21)  $\langle \text{SubDecls} \rangle ::= \langle \text{ProcDecl} \rangle \langle \text{SubDecls} \rangle$
- 22)  $\langle \text{SubDecls} \rangle ::= \epsilon$
  
- 23)  $\langle \text{FunDecl} \rangle ::= \text{KW\_FUNCTION TK\_IDENT} \langle \text{Params} \rangle \text{SB\_COLON} \langle \text{BasicType} \rangle \text{SB\_SEMICOLON} \langle \text{Block} \rangle \text{SB\_SEMICOLON}$
  
- 24)  $\langle \text{ProcDecl} \rangle ::= \text{KW\_PROCEDURE TK\_IDENT} \langle \text{Params} \rangle \text{SB\_SEMICOLON} \langle \text{Block} \rangle \text{SB\_SEMICOLON}$
  
- 25)  $\langle \text{Params} \rangle ::= \text{SB\_LPAR} \langle \text{Param} \rangle \langle \text{Params2} \rangle \text{SB\_RPAR}$
- 26)  $\langle \text{Params} \rangle ::= \epsilon$
  
- 27)  $\langle \text{Params2} \rangle ::= \text{SB\_SEMICOLON} \langle \text{Param} \rangle \langle \text{Params2} \rangle$
- 28)  $\langle \text{Params2} \rangle ::= \epsilon$
  
- 29)  $\langle \text{Param} \rangle ::= \text{TK\_IDENT SB\_COLON} \langle \text{BasicType} \rangle$
- 30)  $\langle \text{Param} \rangle ::= \text{KW\_VAR TK\_IDENT SB\_COLON} \langle \text{BasicType} \rangle$

# Văn phạm KPL viết bằng BNF

- 31) `<Type> ::= KW_INTEGER`
- 32) `<Type> ::= KW_CHAR`
- 33) `<Type> ::= TK_IDENT`
- 34) `<Type> ::= KW_ARRAY SB_LSEL TK_NUMBER SB_RSEL KW_OF <Type>`
  
- 35) `<BasicType> ::= KW_INTEGER`
- 36) `<BasicType> ::= KW_CHAR`
  
- 37) `<UnsignedConstant> ::= TK_NUMBER`
- 38) `<UnsignedConstant> ::= TK_IDENT`
- 39) `<UnsignedConstant> ::= TK_CHAR`
  
- 40) `<Constant> ::= SB_PLUS <Constant2>`
- 41) `<Constant> ::= SB_MINUS <Constant2>`
- 42) `<Constant> ::= <Constant2>`
- 43) `<Constant> ::= TK_CHAR`
  
- 44) `<Constant2> ::= TK_IDENT`
- 45) `<Constant2> ::= TK_NUMBER`
  
- 46) `<Statements> ::= <Statement> <Statements2>`
  
- 47) `<Statements2> ::= SB_SEMICOLON <Statement> <Statements2>`
- 48) `<Statements2> ::= ε`

# Văn phạm KPL viết bằng BNF

- 49) `<Statement> ::= <AssignSt>`
- 50) `<Statement> ::= <CallSt>`
- 51) `<Statement> ::= <GroupSt>`
- 52) `<Statement> ::= <IfSt>`
- 53) `<Statement> ::= <WhileSt>`
- 54) `<Statement> ::= <ForSt>`
- 55) `<Statement> ::=  $\epsilon$`
  
- 56) `<AssignSt> ::= <Variable> SB_ASSIGN <Expression>`
- 57) `<AssignSt> ::= TK_IDENT SB_ASSIGN <Expression>`
  
- 58) `<CallSt> ::= KW_CALL TK_IDENT <Arguments>`
  
- 59) `<GroupSt> ::= KW_BEGIN <Statements> KW_END`
  
- 60) `<IfSt> ::= KW_IF <Condition> KW_THEN <Statement> <ElseSt>`
  
- 61) `<ElseSt> ::= KW_ELSE <Statement>`
- 62) `<ElseSt> ::=  $\epsilon$`
  
- 63) `<WhileSt> ::= KW_WHILE <Condition> KW_DO <Statement>`
- 64) `<ForSt> ::= KW_FOR TK_IDENT SB_ASSIGN <Expression> KW_TO  
                  <Expression> KW_DO <Statement>`

# Văn phạm KPL viết bằng BNF

65)  $\langle \text{Arguments} \rangle ::= \text{SB\_LPAR } \langle \text{Expression} \rangle \langle \text{Arguments2} \rangle \text{SB\_RPAR}$

66)  $\langle \text{Arguments} \rangle ::= \varepsilon$

67)  $\langle \text{Arguments2} \rangle ::= \text{SB\_COMMA } \langle \text{Expression} \rangle \langle \text{Arguments2} \rangle$

68)  $\langle \text{Arguments2} \rangle ::= \varepsilon$

68)  $\langle \text{Condition} \rangle ::= \langle \text{Expression} \rangle \langle \text{Condition2} \rangle$

69)  $\langle \text{Condition2} \rangle ::= \text{SB\_EQ } \langle \text{Expression} \rangle$

70)  $\langle \text{Condition2} \rangle ::= \text{SB\_NEQ } \langle \text{Expression} \rangle$

71)  $\langle \text{Condition2} \rangle ::= \text{SB\_LE } \langle \text{Expression} \rangle$

72)  $\langle \text{Condition2} \rangle ::= \text{SB\_LT } \langle \text{Expression} \rangle$

73)  $\langle \text{Condition2} \rangle ::= \text{SB\_GE } \langle \text{Expression} \rangle$

74)  $\langle \text{Condition2} \rangle ::= \text{SB\_GT } \langle \text{Expression} \rangle$

75)  $\langle \text{Expression} \rangle ::= \text{SB\_PLUS } \langle \text{Expression2} \rangle$

76)  $\langle \text{Expression} \rangle ::= \text{SB\_MINUS } \langle \text{Expression2} \rangle$

77)  $\langle \text{Expression} \rangle ::= \langle \text{Expression2} \rangle$

78)  $\langle \text{Expression2} \rangle ::= \langle \text{Term} \rangle \langle \text{Expression3} \rangle$

79)  $\langle \text{Expression3} \rangle ::= \text{SB\_PLUS } \langle \text{Term} \rangle \langle \text{Expression3} \rangle$

80)  $\langle \text{Expression3} \rangle ::= \text{SB\_MINUS } \langle \text{Term} \rangle \langle \text{Expression3} \rangle$

81)  $\langle \text{Expression3} \rangle ::= \varepsilon$

# Văn phạm KPL viết bằng BNF

- 82)  $\langle \text{Term} \rangle ::= \langle \text{Factor} \rangle \langle \text{Term2} \rangle$
- 83)  $\langle \text{Term2} \rangle ::= \text{SB\_TIMES} \langle \text{Factor} \rangle \langle \text{Term2} \rangle$
- 84)  $\langle \text{Term2} \rangle ::= \text{SB\_SLASH} \langle \text{Factor} \rangle \langle \text{Term2} \rangle$
- 85)  $\langle \text{Term2} \rangle ::= \varepsilon$
- 86)  $\langle \text{Factor} \rangle ::= \langle \text{UnsignedConstant} \rangle$
- 87)  $\langle \text{Factor} \rangle ::= \langle \text{Variable} \rangle$
- 88)  $\langle \text{Factor} \rangle ::= \langle \text{FunctionApptication} \rangle$
- 89)  $\langle \text{Factor} \rangle ::= \text{SB\_LPAR} \langle \text{Expression} \rangle \text{SB\_RPAR}$
- 90)  $\langle \text{Variable} \rangle ::= \text{TK\_IDENT} \langle \text{Indexes} \rangle$
- 91)  $\langle \text{FunctionApplication} \rangle ::= \text{TK\_IDENT} \langle \text{Arguments} \rangle$
- 92)  $\langle \text{Indexes} \rangle ::= \text{SB\_LSEL} \langle \text{Expression} \rangle \text{SB\_RSEL} \langle \text{Indexes} \rangle$
- 93)  $\langle \text{Indexes} \rangle ::= \varepsilon$

- Đọc vào: các hàm
  - ReadI: Đọc 1 số nguyên. Không tham số
  - ReadC: Đọc 1 ký tự. Không tham số

Ví dụ

```
var a: integer;  
a:= ReadI;
```

- In ra: các thủ tục
  - Writel: In ra 1 số nguyên. 1 tham số
  - WriteC: In ra 1 ký tự. 1 tham số
  - WriteLn: In dấu xuống dòng. Không tham số

Ví dụ

```
call Writel(a);  
call WriteLn;
```

# Hàm tính tổng hai số nguyên

```
function sum (x: integer; y:integer): integer;  
begin  
    sum:=x+y;  
end;
```


Chú ý: **Ngôn ngữ KPL không có lệnh return.** Trả kết quả của hàm bằng cách viết lệnh gán với tên hàm ở vế trái.

Bài tập: Viết hai chương trình trên KPL nhập vào hai số nguyên và in ra tổng của chúng, một chương trình nhập và in trực tiếp, một chương trình sử dụng hàm tính tổng hai số nguyên,




- Là công cụ để mô tả cú pháp của ngôn ngữ lập trình dưới dạng đồ thị
- Mỗi sơ đồ cú pháp là một đồ thị định hướng với lối vào và lối ra xác định.
- Mỗi sơ đồ cú pháp có một tên duy nhất

# Ví dụ một sơ đồ cú pháp




# Sơ đồ cú pháp của KPL (Tổng thể CT)


program





# Sơ đồ cú pháp của KPL (Khối)




# Sơ đồ cú pháp của KPL (Khai báo)




# Sơ đồ cú pháp của KPL(tham số, hằng không dấu)



# Sơ đồ cú pháp của KPL (lệnh)



# Sơ đồ cú pháp của KPL (biểu thức)





# Sơ đồ cú pháp của KPL (thừa số, điều kiện)

